Функции от по-висок ред

Трифон Трифонов

Функционално програмиране, 2018/19 г.

17 октомври 2018 г.

Подаване на функции като параметри

В Scheme функциите са “първокласни” стойности.

Примери:

- `(define (fixed-point? f x) (= (f x) x))`
- `(fixed-point? sin 0)` ⟶ `#t`
- `(fixed-point? exp 1)` ⟶ `#f`
- `(fixed-point? expt 0)` ⟶ Грешка!
- `(define (branch p? f g x) ((if (p? x) f g) x))`
- `(branch odd? exp fact 4)` ⟶ 24
- `(define (id x) x)`
- `(branch number? log id "1")` ⟶ `"1"`
- `(branch string? number? procedure? symbol?)` ⟶ `#t`

Функции от по-висок ред

Дефиниция

Функция, която приема функция за параметър се нарича *функция от по-висок ред*.

- `fixed-point?` и `branch` са функции от по-висок ред
- Примери за математически функции от по-висок ред?
- Всички функции в λ-смятането са от по-висок ред!

Задачи за сумиране

Задача: Да се пресметнат следните суми:

1. $k^2 + (k+1)^2 + \ldots + 100^2$ за $k \leq 100$
2. $\int_a^b f(x)dx \approx \Delta x\left[f(a) + f(a+\Delta x) + f(a+2\Delta x) + \ldots + f(b)\right]$
3. $x + e^x + e^{e^x} + e^{e^{e^x}} + \ldots$ докато поредното събираемо е $\leq 10^{1000}$

```
(define (sum1 k)
  (if (> k 100) 0 (+ (* k k) (sum1 (+ k 1)))))

(define (sum2 a b f dx)
  (if (> a b) 0 (+ (* dx (f a)) (sum2 (+ a dx) b f dx))))

(define (sum3 x)
  (if (> x (expt 10 1000)) 0 (+ x (sum3 (exp x)))))
```

Обобщена функция за сумиране

Да се напише функция от по-висок ред `sum`, която пресмята сумата:

$$\sum_{\substack{i=a \\ i \leftarrow next(i)}}^{b} term(i).$$

```
(define (sum a b term next)
  (if (> a b) 0 (+ (term a) (sum (next a) b term next))))
```

Приложения на sum

Решение на задачите за суми чрез sum:

$$\sum_{i=k}^{100} i^2$$

```
(define (square x) (* x x))
(define (1+ x) (+ x 1))
(define (sum1 k) (sum k 100 square 1+))
```

$$\Delta x \sum_{\substack{i=a \\ i \to i+\Delta x}}^{b} \Delta x\, f(i)$$

```
(define (sum2 a b f dx)
  (define (term x) (* dx (f x)))
  (define (next x) (+ x dx))
  (* dx (sum a b term next)))
```

$$\sum_{\substack{i=x \\ i \to e^i}}^{10^{1000}} i$$

```
(define (sum3 x)
(sum x (expt 10 1000) id exp))
```

Обобщена функция за произведение

Да се напише функция от по-висок ред `product`, която пресмята:

$$\prod_{\substack{i=a\\ i\leftarrow next(i)}}^{b} term(i).$$

```
(define (prod a b term next)
  (if (> a b) 1 (* (term a) (prod (next a) b term next))))

(define (sum a b term next)
  (if (> a b) 0 (+ (term a) (sum (next a) b term next))))
```

Обобщена функция за натрупване

Да се напише функция, която пресмята

$$term(a) \oplus \Big(term\big(next(a)\big) \oplus \big(\ldots \oplus (term(b) \oplus \perp) \ldots\big)\Big),$$

където $\oplus$ е бинарна операция,
а $\perp$ е нейната "нулева стойност", т.е. $x \oplus \perp = x$.

```
(define (accumulate op nv a b term next)
  (if (> a b) nv
      (op (term a) (accumulate op nv (next a) b term next))))

(define (sum a b term next) (accumulate + 0 a b term next))
(define (product a b term next) (accumulate * 1 a b term next))
```

Задача: пресмятане на полином

Да се пресметне стойността на полинома

$$
\begin{aligned}
P_n(x) &= x^n + 2x^{n-1} + \ldots + (n-2)x^3 + (n-1)x^2 + nx + (n+1) \\
&= \sum_{i=0}^{n} (n+1-i)x^i
\end{aligned}
$$

Решение №1:

```
(define (p n x)
  (define (term i) (* (- (1+ n) i) (expt x i)))
  (accumulate + 0 0 n term 1+))
```

Можем ли да решим задачата без да извикваме expt на всяка стъпка?

Правило на Хорнер

$$\begin{aligned}P_n(x) &= x^n + 2x^{n-1} + \ldots + (n-2)x^3 + (n-1)x^2 + nx + (n+1)\\ &= \Bigg(\bigg(\Big(\ldots((x+2)x+3)x+\ldots\Big)x+(n-1)\bigg)x+n\Bigg)x+(n+1)\end{aligned}$$

Можем ли да сметнем с accumulate?

Идея: Да използваме операцията $u \oplus v := ux + v$.

Коя е "нулевата стойност" $\perp$?

Решение №2:

```
(define (p n x)
  (define (op u v) (+ (* u x) v))
  (accumulate op 0 1 (1+ n) id 1+))
```

Не смята правилно!

Правило на Хорнер

Всъщност пресметнахме:

$$Q_n(x) = x + 2x + 3x + \ldots + nx + (n+1)x = \frac{(n+1)(n+2)}{2}x.$$

Идея: Да използваме операцията $u \oplus v := u + vx$.

Решение №3:

```
(define (p n x)
  (define (op u v) (+ u (* v x)))
  (accumulate op 0 1 (1+ n) id 1+))
```

Пак не смята правилно!!!

Ляво и дясно натрупване

Всъщност пресметнахме:

$$\begin{aligned} R_n(x) &= 1 + x\Bigg(2 + x\Bigg(\ldots + x\Bigg((n-1) + x(n + x(n+1))\Bigg)\ldots\Bigg)\Bigg) \\ &= (n+1)x^n + nx^{n-1} + (n-1)x^{n-2} + \ldots + 3x^2 + 2x + 1 \end{aligned}$$

вместо

$$\begin{aligned} P_n(x) &= \Bigg(\Bigg(\Bigg(\ldots((x+2)x+3)x+\ldots\Bigg)x + (n-1)\Bigg)x + n\Bigg)x + (n+1) \\ &= x^n + 2x^{n-1} + \ldots + (n-2)x^3 + (n-1)x^2 + nx + (n+1). \end{aligned}$$

За неасоциативни операции $\oplus$ има значение в какъв ред са скобите!

Обобщена функция за ляво натрупване

Да се напише функция, която пресмята **ляво натрупване**:

$$\Big(\dots\big((\perp \oplus term(a)) \oplus term(next(a))\big) \oplus \dots\Big) \oplus term(b)$$

```
(define (accumulate-i op nv a b term next)
  (if (> a b) nv
      (accumulate-i op (op nv (term a)) (next a) b term next)))
```

- accumulate — дясно натрупване, рекурсивен процес
- accumulate-i — ляво натрупване, итеративен процес

Правило на Хорнер

$$
\begin{aligned}
P_n(x) &= x^n + 2x^{n-1} + \ldots + (n-2)x^3 + (n-1)x^2 + nx + (n+1) \\
&= \Bigg(\bigg(\Big(\ldots((x+2)x+3)x+\ldots\Big)x+(n-1)\bigg)x+n\Bigg)x+(n+1)
\end{aligned}
$$

Идея: използваме `accumulate-i` и $u \oplus v := ux + v$.

Решение №4:

```
(define (p n x)
  (define (op u v) (+ (* u x) v))
  (accumulate-i op 0 1 (1+ n) id 1+))
```

Анонимни функции

Можем да конструираме параметрите на функциите от по-висок ред "на място", без да им даваме имена!

- (lambda ({<параметър>}) <тяло>)
- Оценява се до функционален обект със съответните параметри и тяло
- Анонимната функция пази указател към средата, в която е оценена
- Примери:
  - `(lambda (x) (+ x 3))` ⟶ `#<procedure>`
  - `((lambda (x) (+ x 3)) 5)` ⟶ `8`
  - `(define (<име> <параметри>) <тяло>)`

    ⟺

    `(define <име> (lambda (<параметри>) <тяло>))`

Примери

```
(define (integral a b f dx)
   (* dx (accumulate + 0 a b f (lambda (x) (+ x dx)))))

(define (p n x)
   (accumulate-i (lambda (u v) (+ (* u x) v)) 0
                 1 (+ n 1) (lambda (i) (+ i 1))))
```

Задача: Как можем да реализираме с accumulate:

- $n!$
- x^n
- $\sum_{i=0}^{n} \frac{x^i}{i!}$
- $\exists x \in [a; b]\ p(x)$

Функции, които връщат функции

Да разгледаме функция, която прилага дадена функция два пъти над аргумент.

- `(define (twice f x) (f (f x)))`
- `(twice square 3)` ⟶ 81
- `(define (twice f) (lambda (x) (f (f x))))`
- `(twice square 3)` ⟶ Грешка!
- `(twice square)` ⟶ `#<procedure>`
- `((twice square) 3)` ⟶ 81
- `((twice (twice square)) 2)` ⟶ 65536

Примери

- `(define (n+ n) (lambda (i) (+ i n)))`
- `(define 1+ (n+ 1))`
- `(1+ 7)` ⟶ 8
- `(define 5+ (n+ 5))`
- `(5+ 7)` ⟶ 12
- `(define (compose f g) (lambda (x) (f (g x))))`
- `((compose square 1+) 3)` ⟶ 16
- `((compose 1+ square) 3)` ⟶ 10
- `((compose 1+ (compose square (n+ 2))) 3)` ⟶ 26

Оценка на lambda

{**E**}	`(define (n+ n) (lambda (i) (+ i n)))`
{**E**}	`(define 1+ (n+ 1))`
{**E**}	`(define 5+ (n+ 5))`
{**E**}	`(1+ 7)`
	↓
{$\mathbf{E}_3$}	`(+ i n)`
	↓
	8
{**E**}	`(5+ 7)`
	↓
{$\mathbf{E}_4$}	`(+ i n)`
	↓
	12



Намиране на производна

$$f'(x) = \lim_{\Delta x \to 0} \frac{f(x+\Delta x) - f(x)}{\Delta x}$$

```
(define (derive f dx)
  (lambda (x) (/ (- (f (+ x dx)) (f x)) dx)))
```

- `(define 2* (derive square 0.01))`
- `(2* 5)` ⟶ `10.009999999999764`
- `((derive square 0.0000001) 5)` ⟶ `10.000000116860974`
- `((derive (derive (lambda (x) (* x x x)) 0.001) 0.001) 3)` ⟶ `18.006000004788802`

Повторено прилагане

Да се намери n-кратното прилагане на дадена едноместна функция.

$$f^n(x) = \underbrace{f(f(f(\ldots(f}_{n}(x))\ldots)))$$

Решение №1: $f^0(x) = x, f^n(x) = f(f^{n-1}(x))$

```
(define (repeated f n)
  (lambda (x) (if (= n 0) x (f ((repeated f (- n 1)) x)))))
```

Решение №2: $f^0 = id, f^n = f \circ f^{n-1}$

```
(define (repeated f n)
  (if (= n 0) id (compose f (repeated f (- n 1)))))
```

Решение №3: $f^n = \underbrace{f \circ f \circ \ldots \circ f}_{n} \circ id$

```
(define (repeated f n)
  (accumulate compose id 1 n (lambda (i) f) 1+))
```

n-та производна

Да се намери n-та производна на дадена едноместна функция.

Решение №1: $f^{(0)} = f, f^{(n)} = (f^{(n-1)})'$

```
(define (derive-n f n dx)
  (if (= n 0) f (derive (derive-n f (- n 1) dx) dx)))
```

Решение №2: $f^{(n)} = f^{\overbrace{''''\ldots''}^{n}}$

```
(define (derive-n f n dx)
  ((repeated (lambda (f) (derive f dx)) n) f))
```